UCB Math 124, Spring 2023: Final Exam

Prof. Persson, May 11, 2023

SOLUTIONS

Name: _____

SID: _____

Instructions:

- One sheet of notes, no books, no calculators.
- Exam time 180 minutes, do all of the problems.
- You must justify your answers for full credit.
- All computer codes must be written in the Julia programming language, using only the functionality and the packages covered in the course.
- Write your answers next to or below each problem.
- If you need more space, use reverse side or scratch pages.
  Indicate clearly where to find your answers.

1. (6 points) Find the asymptotic operation counts and memory usage for the Julia functions below, using Big-O notation with brief justifications.

a)

```
x = rand(n)
s = 0.0
for i = 1:n, j = i+1:n
    s += abs(x[j] - x[i])
end
```

Solution: Operations: first loop n iterations, second $n - i$, constant operations per iteration, gives total $\mathcal{O}(n^2)$. Memory: $\mathcal{O}(n)$ since one array of length n.

b)

```
s = 0.0
for i = 1:n, j = 2:2:n
    k = n
    while k > 0
        k ÷= 3
        s += 1 / (i + j + k)
    end
end
```

Solution: Operations: first loop n iterations, second $n/2$, while loop $\lceil \log_3 n \rceil$ iterations (see HW7), gives total $\mathcal{O}(n^2 \log n)$. Memory: $\mathcal{O}(1)$ since no arrays.

c)

```
x = Float64[]
for i = 1:n
    for j = 1:n
        if j > 1000
            break
        end
        newx = randn()
        if j % 10 == 0
            push!(x, newx)
        end
    end
end
```

Solution: Operations: first loop n iterations, second at most 1000 independently of n, constant operations per iteration, gives total $\mathcal{O}(n)$ operations. Memory: $\mathcal{O}(n)$ since the array `x` will get at most $1000/10 = 100$ new entries for each value of i, independently of n.

2. Consider the following game: Given an integer N, random integers $x \in 1, 2, \ldots, N-1$ are drawn (with replacement) until the sum of any two numbers equals N. The "score" is the number of drawn numbers D. For example, if $N = 20$ and you get these numbers:

$$5, 19, 9, 8, 3, 12$$

then the score $D = 6$ since $8 + 12 = 20$.

a) (3 points) Write a Julia function `sumgame(N)` which simulates one instance of the game and returns the score D. The operation count for your code must be $\mathcal{O}(D)$ and the memory usage $\mathcal{O}(N)$. *Hint*: Use a boolean array to record the numbers.

b) (1 points) Write a Julia function `average_score(N,ntrials)` which uses Monte Carlo simulation with `ntrials` trials to estimate the expected value of the score.

Solution:

```
function sumgame(N)
    found = falses(N-1)
    D = 0
    while true
        D += 1
        x = rand(1:N-1)
        if found[N-x]
            return D
        end
        found[x] = true
    end
end
```

```
function average_score(N, ntrials)
    total = 0
    for i = 1:ntrials
        total += sumgame(N)
    end
    return total / ntrials
end
```

3. If p is the perimeter of a right angle triangle with integer length sides, $\{a, b, c\}$, with $a \leq b \leq c$, there are exactly $n = 3$ solutions for $p = 120$:

$$\{20, 48, 52\}, \{24, 45, 51\}, \{30, 40, 50\}$$

a) (3 points) Write a Julia function `integer_triangles(p)` which returns the number of solutions n for the perimeter p, using operation count $\mathcal{O}(p^2)$ and memory $\mathcal{O}(1)$.

b) (1 points) Write a *one-line* Julia code which finds the value of $p \leq 1000$ for which n is maximized (use `argmax` to find the index of the largest element of an array).

Solution:

```
function integer_triangles(p)
    count = 0
    for a = 1:p
        for b = a:p
            c = p - a - b
            if a^2 + b^2 == c^2
                count += 1
            end
        end
    end
    count
end

argmax(integer_triangles.(1:1000))
```

4. (4 points) Consider the n-by-n matrix defined as follows, where n is an odd integer: Start in the central column of the first row with the number 1. Next, insert increasing numbers by moving diagonally up and right one step at a time. When an "up and to the right" move would leave the square, it is wrapped around to the last row or first column, respectively. If a filled square is encountered, move vertically down one square instead, again wrapping around, then continue as before. See below for an example when $n = 5$.

$$\begin{pmatrix} 17 & 24 & 1 & 8 & 15 \\ 23 & 5 & 7 & 14 & 16 \\ 4 & 6 & 13 & 20 & 22 \\ 10 & 12 & 19 & 21 & 3 \\ 11 & 18 & 25 & 2 & 9 \end{pmatrix}$$

Write a Julia function `mysquare(n)` which creates and returns this matrix for size n.

Solution: One difficulty is the "wrapping around" for indices outside the range $1, \ldots, n$. In the solution below, this is handled conveniently with a helper function `wrap`.

```
function mysquare(n)
    A = zeros(Int64,n,n)
    i,j = 1,(n+1)÷2
    wrap(i) = mod(i-1,n) + 1
    for nbr = 1:n^2
        A[i,j] = nbr
        if A[wrap(i-1),wrap(j+1)] == 0
            i = wrap(i-1)
            j = wrap(j+1)
        else
            i = wrap(i+1)
        end
    end
    return A
end
```

5. (4 points) The shape in the picture is composed of the following three parametric functions:

$$(x, y) = (\cos t, \sin t), \quad 0 \le t \le \pi/2$$
$$(x, y) = (-\sqrt{2}\sin(3\pi t/4), 1 - t), \quad 0 \le t \le 1$$
$$(x, y) = (t, t^4 - 1), \quad -1 \le t \le 1$$



Consider a triangle with one vertex on each of these three curves. Use the `Optim` package, with the `Newton()` solver and the `autodiff=:forward` option, to find the triangle with the maximum area. Use the midpoints of the parameter range for each curve as initial guesses. You can assume that the parameters will remain within their ranges. Implement your code in a function `max_tri_area()` which returns a vector with the three x, y-coordinates of the optimal triangle.

Solution:

```
using Optim

function max_tri_area()
    xy1(t) = [cos(t), sin(t)]
    xy2(t) = [-sqrt(2)*sin(3πt/4), 1-t]
    xy3(t) = [t, t^4-1]
    tinit = [π/4, 0.5, 0.0]
    allxy(ts) = [xy1(ts[1]), xy2(ts[2]), xy3(ts[3])]

    function tri_area(tri)
        d12 = tri[2] - tri[1]
        d13 = tri[3] - tri[1]
        area = abs(d12[1] * d13[2] - d12[2] * d13[1]) / 2
    end
    fobj(ts) = -tri_area(allxy(ts))

    res = optimize(fobj, tinit, Newton(); autodiff=:forward)
    display(res)
    return allxy(res.minimizer)
end
```